
*** Theory EQ* of equality predicate
*** plus the choice of an element * of sort Elt.

fth EQ* is protecting BOOL .  
  sort Elt .
  op * : -> Elt .           *** chosen element in Elt 
  op eq : Elt Elt -> Bool . *** equality predicate

  vars X Y Z : Elt .

  eq eq(X,X) = true .
  ceq X = Y if eq(X,Y) = true [nonexec] .
endfth

*** CINNI is Mark-Oliver Stehr's calculus of explicit
*** substitution for any binding operator; here is the
*** lambda calculus case.

mod LAMBDA-CINNI{N :: EQ*} is protecting NAT .
  sorts Lambda{N} Sub{N} .

  op _{_} : N$Elt Nat ->  Lambda{N} [ctor] .

  op \_._ : N$Elt Lambda{N} -> Lambda{N} [ctor] .
                            *** lambda abstraction
                            
  op __ : Lambda{N} Lambda{N} -> Lambda{N}
             [ctor gather (E e)] . *** application
             
  op if : Bool Lambda{N} Lambda{N} -> Lambda{N} .
                                *** if-then-else

  op [_:=_] : N$Elt Lambda{N} -> Sub{N} [ctor] .
                                *** substitution



                                
  op [^_] : N$Elt -> Sub{N} [ctor] .  *** shift

  op [^__] : N$Elt Sub{N} -> Sub{N} [ctor] .
                                     *** lift
                                     
  op __ : Sub{N} Lambda{N} -> Lambda{N} .
                     *** substitution application

  vars X Y : N$Elt . vars U V : Lambda{N} .
  var n : Nat .  var SU : Sub{N} .

  eq if(true,V,U) = V .       eq if(false,V,U) = U .

  eq [1] :  [X := U] (Y{0}) = if (eq(X,Y),U,Y{0}) .
  eq [2] :  [X := U] (Y{s(n)}) =
                         if(eq(X,Y),Y{n},Y{s(n)}) .

  eq [3] : [^ X] (Y{n}) = if(eq(X,Y),Y{s(n)},Y{n}) .

  eq [4] : [^ X SU] (Y{0}) =
               if(eq(X,Y),Y{0},[^ X] (SU (Y{0}))) .
  eq [5] : [^ X SU] (Y{s(n)}) =
      if(eq(X,Y),[^ Y] (SU (Y{n})),[^ X] (SU (Y{s(n)}))) .

  eq [6] : SU (U V) = (SU U) (SU V) .
  eq [7] : SU (\ X . U) = \ X . ([^ X SU] U) .

  rl [beta] : (\ X . U) V => [X := V] U  .  
endm

view Qid from EQ* to QID is
sort Elt to Qid .
op * to term 'x .
vars X Y : Elt .
op eq(X,Y) to term (X == Y) .



endv

mod LAMBDA-CINNI-BETA is
protecting LAMBDA-CINNI{Qid} .

ops I K S ff : -> Lambda{Qid} .
eq I = \ 'x . 'x{0} .
eq K = \ 'x . \ ('y) . 'x{0} .
eq ff = \ 'x . \ ('y) . (('y){0}) .
eq S = \ 'x . \ ('y) . \ ('z) .
         ((('x{0}) (('z){0})) ((('y){0}) (('z){0}))) .

  var X : Qid .  var U : Lambda{Qid} .
endm

rew K .

rew ff .

rew S .

*** Consider the beta reduction
***
***( (\ 'x . ('x \ 'y .(('x 'y) 'z))) ('v 'v) )
***
***( ->beta ('v 'v) \ 'y .((('v 'v) 'y) 'z) )
***
*** in CINNI notation should be:
***
***( (\ 'x .('x{0} \ 'y .(('x{0} 'y{0}) 'z{0})))
                                    ('v{0} 'v{0}) )
***
***( ->beta 'v{0} ('v{0})
         (\ 'y . ('v{0} ('v{0}) ('y{0}) ('z{0}))) )

rew (\ 'x .('x{0} (\ ('y) .(('x{0} (('y){0}))



             (('z){0}))))) ((('v){0}) (('v){0}))  .

*** Consider the "weird renaming" situation
***
***( (\ 'x . \ 'y . x) 'y ->beta \ 'z . 'y )
***
*** in CINNI notation should get
***
***( (\ 'x . \ 'y . ('x{0})) (('y){0}) ->beta \ 'y . 'y{1}
)
***

rew (\ 'x . (\ 'y . ('x{0}))) (('y){0}) .

*** Let us execute Cornell's example:
***( (\ 'z . \ 'v . ('z x)) 'y ->beta \ 'v . ('y x) )
***
*** in CINNI notation this becomes:
***
***( (\ 'z . (\ 'v . (('z){0} ('x{0})))) (('y){0})
  ->beta  \ 'v . ('y{0} ('x{0}))
)

rew (\ 'z . (\ 'v . (('z){0} ('x{0})))) (('y){0}) .

*** for another example, consider the "avoidance of
variable
*** capture" problem in beta-reducing the expression:
***
***(  (\ x . \ y . (y x)) (\ x . (x y)) -->beta
    \z . (z \ x . (x y))

where a fresh variable 'z' had to be chosen.
)
*** In CIINI notation this becomes the reduction:



***
***(  (\ 'x . (\ 'y . ('y{0} ('x{0})))) (\ 'x . ('x{0}
('y{0})))
      -->beta \ 'y . ('y{0} (\ 'x . ('x{0} ('y{1}))))

where the need for fresh names evaporates: y{0} is "jacked
up"
to remain a free variable y{1} under the \ 'y
)

rew ((\ 'x . (\ 'y . ('y{0} ('x{0})))) (\ 'x . ('x{0}
('y{0})))) .

*** Alpha conversion can now be added to get a quotient of
*** CINNI which, for CLOSED terms, is ISOMORPHIC to
*** the subsest of closed terms in LAMBDA-DB.  
*** The definition of [alpha] could have been added to the
*** above parameterized module LAMBDA-CINNI{N :: EQ*}
*** and then the [alpha] equation would have been defined 
*** as follows:
***
***   var X : N$Elt . var U : Lambda{N} .
***
*** ceq [alpha] : \ X . U  =  \ * . ([X := *{0}] [^ *] U)
***                                  if eq(X,*) = false .
***
*** We would be free to CHOOSE any name we wish for * in 
*** instances of LAMBDA-CINNI{N :: EQ*} by using a view.
***
*** But to FIRST show how beta ALONE works, [alpha] was NOT
*** added to LAMBDA-CINNI{N :: EQ*} .  However, we can do 
***  it now in the following extension
*** LAMBDA-CINNI-BETA-ALPHA of LAMBDA-CINNI-BETA where:
***
*** 1. we choose the variable 'x as the interpretation of
***    '*' and



*** 2. the generic eq(X,*) is here replaced by (X == 'x)

mod LAMBDA-CINNI-BETA-ALPHA is
protecting LAMBDA-CINNI-BETA .

  var X : Qid .  var U : Lambda{Qid} .

  ceq [alpha] : \ X . U  =  \ 'x . ([X := 'x{0}] [^ 'x] U)
                                  if (X == 'x) = false .

endm

*** Now for K we get a notation isomorphic to De Bruijn's:
***
*** \ 'x . \ 'x . 'x{1}

rew K .

*** Likewise ff (second projection) becomes: \ 'x . \ 'x .
'x{0}

rew ff .

*** And S becomes:
***( \ 'x . \ 'x . \ 'x . ('x{2} ('x{0}) ('x{1} ('x{0}))) )

rew S .

*** Consider the beta reduction
***
***( (\ 'x . ('x \ 'y .(('x 'y) 'z))) ('v 'v) )
***
***( ->beta ('v 'v) \ 'y .((('v 'v) 'y) 'z) )
***
*** where 'z and 'v are FREE VARIABLES 



***
*** in CINNI notation but using [alpha] should be:
***
***( ((\ 'x .('x{0} \ 'y .(('x{0} 'y{0}) 'z{0})))
                                    ('v{0} 'v{0}) )
    ->beta+alpha 'v{0} ('v{0})
     (\ 'x . ('v{0} ('v{0}) ('x{0}) ('z{0})))
)
*** note that [alpha] does NOT change the name of
*** the free variables 'z and 'v .  That is, only
*** terms without free variables have a UNIQUE 
*** representative up to alpha-conversion, using only 
*** the name x thanks to the equation [alpha]

rew (\ 'x .('x{0} (\ ('y) .(('x{0} (('y){0}))
             (('z){0}))))) ((('v){0}) (('v){0}))  .

*** Consider the "weird renaming" situation
***
***( (\ 'x . \ 'y . x) 'y ->beta \ 'z . 'y   )
***
*** in CINNI notation but using [alpha] should get
***
***( (\ 'x . \ 'y . ('x{0})) (('y){0}) ->beta+alpha \ 'x .
'y{0} )
***

rew (\ 'x . (\ 'y . ('x{0}))) (('y){0}) .

*** Let us execute Cornell's example:
***( (\ 'z . \ 'v . ('z x)) 'y ->beta \ 'v . ('y x) )
***
*** in CINNI notation but using [alpha] this becomes:
***



***( (\ 'z . (\ 'v . (('z){0} ('x{0})))) (('y){0})
  ->beta+alpha \ 'x . ('y{0} ('x{1}))
)

rew (\ 'z . (\ 'v . (('z){0} ('x{0})))) (('y){0}) .


